Computer Sciences Department
The University of Wisconsin
1210 West Dayton Street
Madison, Wisconsin 53706

NUMERICAL STUDIES OF STEADY, VISCOUS, INCOMPRESSIBLE FLOW BETWEEN TWO ROTATING SPHERES

by

Donald Greenspan

Technical Report #120

May 1971

NUMERICAL STUDIES OF STEADY, VISCOUS, INCOMPRESSIBLE FLOW BETWEEN TWO ROTATING SPHERES

by

Donald Greenspan

ABSTRACT

A new numerical method is developed for the solution of steady state, viscous, incompressible flow between two rotating spheres. The Navier-Stokes equations are approximated by a triple sequence of linear problems, each of which has a diagonally dominant coefficient matrix. Examples are given for Reynolds numbers in the range $10 \leq R \leq 10^4$ and for various angular velocity combinations. For appropriate choices of smoothing parameters, the method is exceptionally fast.

1. Introduction

Fluid motions inside rotating containers are of wide interest, as in the studies of gyroscopes, centrifuges, and atmospheric and oceanic circulations (see, e.g., refs.[1],[2],[4]-[11],[13], and the references contained therein). In this paper we will study numerically the steady, nonlinear motion of a viscous, incompressible fluid within a spherical annulus. It will be assumed that the boundary spheres rotate coaxially and that the fluid motion is rotationally symmetric. Analytically such problems have been studied by means of linearization [4],[10],[11], by means of asymptotics [13], and under the assumption that the annulus gap is narrow [4],[10]. Only recently have numerical studies been undertaken on related problems [8],[9].

In XYZ space, then, let S_1 and S_2 be concentric spheres which are centered at the origin, which have respective radii r_1 and r_2, with $r_2 > r_1$, and which both rotate about the Z axis. By the assumption of rotational symmetry, it follows that the motion need be studied only in the plane annular section ABCD, denoted by I, in Figure 1.



Fig. 1

If (x,z) is in I and has "polar" coordinates (r,θ) defined by

$$x = r \sin\theta$$

$$z = r\cos\theta ,$$

and if u, v, w are the velocity components of the fluid at (x,z), where u is in the direction of increasing r, v is in the direction of increasing θ, and w is perpendicular to the meridional plane, then the dimensionless, steady state Navier-Stokes equations of motion are ([4],[8])

$$(1.1) \qquad D^2\psi = M$$

$$(1.2) \qquad D^2\Omega + \frac{R}{r^2\sin\theta}\,[\psi_r\Omega_\theta - \psi_\theta\Omega_r] = 0$$

$$(1.3) \qquad D^2M + R\left\{\frac{2\Omega}{r^3\sin^2\theta}\,[\Omega_\theta \sin\theta - \Omega_r r\cos\theta] + \frac{1}{r^2\sin\theta}\,[\psi_r M_\theta - \psi_\theta M_r] + \frac{2M}{r^3\sin^2\theta}\,[\psi_\theta \sin\theta - \psi_r r\cos\theta]\right\} = 0,$$

where

$$(1.4) \qquad D^2 = \frac{\partial^2}{\partial r^2} + \frac{1}{r^2}\frac{\partial^2}{\partial\theta^2} - \frac{1}{r^2}\cot\theta\frac{\partial}{\partial\theta}$$

(1.5) $u = (\psi_\theta)/(r^2 \sin\ \theta)$

(1.6) $v = -(\psi_r)/(r \sin\ \theta)$

(1.7) $w = \Omega/(r \sin\ \theta)$.

We assume that S_1 has angular velocity ω_1 and S_2 has angular velocity ω_2, so that the following boundary conditions ([4], [8]) are to be satisfied:

(1.8) $\psi = \psi_r = 0$, on arcs AD and BC

(1.9) $\psi = \Omega = M = 0$, on CD

(1.10) $$\Omega = \begin{cases} \omega_1 r_1^2 \sin^2\theta, & \text{on AD} \\ \omega_2 r_2^2 \sin^2\theta, & \text{on BC.} \end{cases}$$

In the next two sections we will show how to extend a numerical method developed for cavity flow problems of arbitrary Reynolds number [3] to the boundary value problem defined by (1.1)-(1.3), (1.8)-(1.10).

2. Difference Equation Approximations

Fundamental to the method to be developed is the approximation of differential equations (1.1)-(1.3) by difference equations which are associated with diagonally dominant linear algebraic systems. This will be accomplished by using a combination of central, forward, and backward difference approximations for derivatives [3] as follows.

System (1.1)-(1.3) can be rewritten as

$$(2.1) \qquad r^2 (\sin\theta)\,\psi_{rr} + (\sin\theta)\,\psi_{\theta\theta} - (\cos\theta)\,\psi_\theta = Mr^2 \sin\theta$$

$$(2.2) \qquad r^2 (\sin\theta)\,\Omega_{rr} + (\sin\theta)\,\Omega_{\theta\theta} + (R\psi_r - \cos\theta)\,\Omega_\theta - R\psi_\theta\Omega_r = 0$$

$$(2.3) \qquad r^2 (\sin^2\theta)\,M_{rr} + (\sin^2\theta)\,M_{\theta\theta} + [R(\sin\theta)\,\psi_r - \sin\theta\cos\theta]\,M_\theta - R\psi_\theta (\sin\theta)\,M_r = R\{\frac{2\Omega}{r}[\Omega_r r\cos\theta - \Omega_\theta \sin\theta] + \frac{2M}{r}[\psi_r r\cos\theta - \psi_\theta\sin\theta]\}.$$

In an r-θ plane, as shown in Figure 2, let the points (r,θ), $(r+\Delta r,\theta)$, $(r,\theta+\Delta\theta)$, $(r-\Delta r,\theta)$, $(r,\theta-\Delta\theta)$ be numbered $0,1,2,3,4$, respectively. Then the differential operators on the left-hand-sides of (2.1)-(2.3) will be approximated by difference operators L_1, L_2, L_3, respectively, by making the following substitutions, where ψ_i, Ω_i, and M_i, $i = 0,1,2,3,4$, are the function values of ψ, Ω, and M, respectively, at the point numbered i. At (r,θ), in each of (2.1) - (2.3), set

$$(2.4)\qquad \psi_{rr} = \frac{\psi_1 - 2\,\psi_0 + \psi_3}{(\Delta r)^2}\,,\quad \psi_{\theta\theta} = \frac{\psi_2 - 2\,\psi_0 + \psi_4}{(\Delta\,\theta)^2}$$

$$(2.5)\qquad \Omega_{rr} = \frac{\Omega_1 - 2\Omega_0 + \Omega_3}{(\Delta r)^2}\,,\ \Omega_{\theta\theta} = \frac{\Omega_2 - 2\Omega_0 + \Omega_4}{(\Delta\theta)^2}$$

$$(2.6)\qquad M_{rr} = \frac{M_1 - 2M_0 + M_3}{(\Delta r)^2}\,,\quad M_{\theta\theta} = \frac{M_2 - 2M_0 + M_4}{(\Delta\theta)^2}\,.$$

In (2.1) set

$$(2.7)\qquad \psi_\theta = \frac{\psi_0 - \psi_4}{\Delta\theta}\,.$$

In (2.2) set

$$(2.8)\qquad \psi_r = \frac{\psi_1 - \psi_3}{2\Delta r}\,,\qquad \psi_\theta = \frac{\psi_2 - \psi_4}{2\Delta\theta}\,,$$

and then choose Ω_θ and Ω_r as follows:

$$(2.9)\qquad \Omega_\theta = \frac{\Omega_2 - \Omega_0}{\Delta\theta}\,,\quad \text{if}\ \left[R\ \frac{\psi_1 - \psi_3}{2\Delta r} - \cos\theta\right] \geq 0$$

$$(2.10)\qquad \Omega_\theta = \frac{\Omega_0 - \Omega_4}{\Delta\theta}\,,\quad \text{if}\ \left[R\ \frac{\psi_1 - \psi_3}{2\Delta r} - \cos\theta\right] < 0$$

$$(2.11)\qquad \Omega_r = \frac{\Omega_1 - \Omega_0}{\Delta r}\,,\quad \text{if}\quad \psi_2 - \psi_4 < 0$$

(2.12) $$\Omega_r = \frac{\Omega_0 - \Omega_3}{\Delta r}, \quad \text{if } \psi_2 - \psi_4 \geq 0.$$

In (2.3) use approximations (2.8) for ψ_r and ψ_θ and then set

(2.13) $$M_\theta = \frac{M_2 - M_0}{\Delta\theta}, \quad \text{if } \left[R\,\frac{\psi_1 - \psi_3}{2\Delta r} - \cos\theta\right] \geq 0$$

(2.14) $$M_\theta = \frac{M_0 - M_4}{\Delta\theta}, \quad \text{if } \left[R\,\frac{\psi_1 - \psi_3}{2\Delta r} - \cos\theta\right] < 0$$

(2.15) $$M_r = \frac{M_1 - M_0}{\Delta r}, \quad \text{if } \psi_2 - \psi_4 < 0$$

(2.16) $$M_r = \frac{M_0 - M_3}{\Delta r}, \quad \text{if } \psi_2 - \psi_4 \geq 0.$$

One thereby obtains at (r, θ) the following difference approximations L_1, L_2, L_3 for the left-hand-sides of (2.1) – (2.3):

(2.17) $$L_1\psi_0 \sim r^2 (\sin\theta)\,\psi_{rr} + (\sin\theta)\,\psi_{\theta\theta} - (\cos\theta)\,\psi_\theta,$$

where

(2.18) $$L_1\psi_0 = \psi_0\left[\frac{-2r^2\sin\theta}{(\Delta r)^2} - \frac{2\sin\theta}{(\Delta\theta)^2} - \frac{\cos\theta}{\Delta\theta}\right] + \psi_1\left[\frac{r^2\sin\theta}{(\Delta r)^2}\right]$$
$$+ \psi_2\left[\frac{\sin\theta}{(\Delta\theta)^2}\right] + \psi_3\left[\frac{r^2\sin\theta}{(\Delta r)^2}\right] + \psi_4\left[\frac{\sin\theta}{(\Delta\theta)^2} + \frac{\cos\theta}{\Delta\theta}\right];$$

(2.19) $$L_2\Omega_0 \sim r^2(\sin\theta)\Omega_{rr} + (\sin\theta)\Omega_{\theta\theta} + \Omega_0(R\psi_r - \cos\theta) - R\psi_\theta\Omega_r,$$

where, if $\left[R\frac{\psi_1 - \psi_3}{2\Delta r} - \cos\theta\right] \geq 0$ and $\psi_2 - \psi_4 \geq 0$, then

(2.20a) $$L_2\Omega_0 = \Omega_0\left[-\frac{2r^2\sin\theta}{(\Delta r)^2} - \frac{2\sin\theta}{(\Delta\theta)^2} - \frac{1}{\Delta\theta}\left(R\frac{\psi_1 - \psi_3}{2\Delta r} - \cos\theta\right) - R\left(\frac{\psi_2 - \psi_4}{2\Delta\theta\Delta r}\right)\right] + \Omega_1\left[\frac{r^2\sin\theta}{(\Delta r)^2}\right] + \Omega_2\left[\frac{\sin\theta}{(\Delta\theta)^2} + \left(R\frac{\psi_1 - \psi_3}{2\Delta r} - \cos\theta\right)\frac{1}{\Delta\theta}\right] + \Omega_3\left[\frac{r^2\sin\theta}{(\Delta r)^2} + \frac{R(\psi_2 - \psi_4)}{2\Delta\theta\Delta r}\right] + \Omega_4\left[\frac{\sin\theta}{(\Delta\theta)^2}\right];$$

if $\left[R\frac{\psi_1 - \psi_3}{2\Delta r} - \cos\theta\right] \geq 0$ and $\psi_2 - \psi_4 < 0$, then

(2.20b) $$L_2\Omega_0 = \Omega_0\left[-\frac{2r^2\sin\theta}{(\Delta r)^2} - \frac{2\sin\theta}{(\Delta\theta)^2} - \left(R\frac{\psi_1 - \psi_3}{2\Delta r} - \cos\theta\right)\frac{1}{\Delta\theta} + R\frac{\psi_2 - \psi_4}{2\Delta\theta\Delta r}\right] + \Omega_1\left[\frac{r^2\sin\theta}{(\Delta r)^2} - R\frac{\psi_2 - \psi_4}{2\Delta r\Delta\theta}\right] + \Omega_2\left[\frac{\sin\theta}{(\Delta\theta)^2} + \frac{1}{\Delta\theta}\left(R\frac{\psi_1 - \psi_3}{2\Delta r} - \cos\theta\right)\right] + \Omega_3\left[\frac{r^2\sin\theta}{(\Delta r)^2}\right] + \Omega_4\left[\frac{\sin\theta}{(\Delta\theta)^2}\right];$$

if $\left[R\frac{\psi_1 - \psi_3}{2\Delta r} - \cos\theta\right] < 0$ and $\psi_2 - \psi_4 \geq 0$, then

(2.20c) $$L_2\Omega_0 = \Omega_0\left[\frac{-2r^2\sin\theta}{(\Delta r)^2} - \frac{2\sin\theta}{(\Delta\theta)^2} + \frac{1}{\Delta\theta}\left(R\frac{\psi_1 - \psi_3}{2\Delta r} - \cos\theta\right) - R\left(\frac{\psi_2 - \psi_4}{2\Delta\theta\Delta r}\right)\right] + \Omega_1\left[\frac{r^2\sin\theta}{(\Delta r)^2}\right] + \Omega_2\left[\frac{\sin\theta}{(\Delta\theta)^2}\right] + \Omega_3\left[\frac{r^2\sin\theta}{(\Delta r)^2} + R\left(\frac{\psi_2 - \psi_4}{2\Delta r\Delta\theta}\right)\right] + \Omega_4\left[\frac{\sin\theta}{(\Delta\theta)^2} - \frac{1}{\Delta\theta}\left(R\frac{\psi_1 - \psi_3}{2\Delta r} - \cos\theta\right)\right];$$

if $\left[R\frac{\psi_1-\psi_3}{2\Delta r}-\cos\theta\right]<0$ and $\psi_2-\psi_4<0$, then

$$(2.20d)\quad L_2\Omega_0=\Omega_0\left[-\frac{2r^2\sin\theta}{(\Delta r)^2}-\frac{2\sin\theta}{(\Delta\theta)^2}+\frac{1}{\Delta\theta}\left(R\frac{\psi_1-\psi_3}{2\Delta r}-\cos\theta\right)\right.$$

$$\left.+R\frac{\psi_2-\psi_4}{2\Delta r\Delta\theta}\right]+\Omega_1\left[\frac{r^2\sin\theta}{(\Delta r)^2}-R\frac{\psi_2-\psi_4}{2\Delta r\Delta\theta}\right]$$

$$+\Omega_2\left[\frac{\sin\theta}{(\Delta\theta)^2}\right]+\Omega_3\left[\frac{r^2\sin\theta}{(\Delta r)^2}\right]+\Omega_4\left[\frac{\sin\theta}{(\Delta\theta)^2}\right.$$

$$\left.-\left(R\frac{\psi_1-\psi_3}{2\Delta r}-\cos\theta\right)\frac{1}{\Delta\theta}\right];$$

and finally,

$$(2.21)\quad L_3M_0\sim r^2(\sin^2\theta)M_{rr}+(\sin^2\theta)M_{\theta\theta}+M_\theta\,[R(\sin\theta)\psi_r$$

$$-\sin\theta\cos\theta]-R\psi_\theta M_r\sin\theta,$$

where, if $\left[R\frac{\psi_1-\psi_3}{2\Delta r}-\cos\theta\right]\geq 0$ and $\psi_2-\psi_4\geq 0$, then

$$(2.22a)\quad L_3M_0=M_0\left[-\frac{2r^2\sin^2\theta}{(\Delta r)^2}-\frac{2\sin^2\theta}{(\Delta\theta)^2}-\left(\frac{1}{\Delta\theta}\right)\left(R\sin\theta\,\frac{\psi_1-\psi_3}{2\Delta r}\right.\right.$$

$$\left.\left.-\sin\theta\cos\theta\right)-R\left(\frac{\psi_2-\psi_4}{2\Delta r\Delta\theta}\right)\sin\theta\right]$$

$$+M_1\left[\frac{r^2\sin^2\theta}{(\Delta r)^2}\right]+M_2\left[\frac{\sin^2\theta}{(\Delta\theta)^2}+\frac{1}{(\Delta\theta)}\left(R\sin\theta\,\frac{\psi_1-\psi_3}{2\Delta r}\right.\right.$$

$$\left.\left.-\sin\theta\cos\theta\right)\right]+M_3\left[\frac{r^2\sin^2\theta}{(\Delta r)^2}+R\left(\frac{\psi_2-\psi_4}{2\Delta r\Delta\theta}\right)\sin\theta\right]$$

$$+M_4\left[\frac{\sin^2\theta}{(\Delta\theta)^2}\right];$$

if $\left[R \frac{\psi_1 - \psi_3}{2\Delta r} - \cos\theta\right] \geq 0$ and $\psi_2 - \psi_4 < 0$, then

(2.22b) $$L_3 M_0 = M_0 \left[\frac{-2r^2 \sin^2\theta}{(\Delta r)^2} - \frac{2\sin^2\theta}{(\Delta\theta)^2} - \frac{1}{\Delta\theta}\left(R \sin\theta \frac{\psi_1 - \psi_3}{2\Delta r} - \sin\theta\cos\theta\right) + \frac{R(\psi_2 - \psi_4)}{2\Delta r \Delta\theta} \sin\theta\right] + M_1\left[\frac{r^2 \sin^2\theta}{(\Delta r)^2} - R\left(\frac{\psi_2 - \psi_4}{2\Delta r\Delta\theta}\right)\sin\theta\right] + M_2\left[\frac{\sin^2\theta}{(\Delta\theta)^2} + \frac{1}{\Delta\theta}\left(R\sin\theta\frac{\psi_1 - \psi_3}{2\Delta r} - \sin\theta\cos\theta\right)\right] + M_3\left[\frac{r^2\sin^2\theta}{(\Delta r)^2}\right] + M_4\left[\frac{\sin^2\theta}{(\Delta\theta)^2}\right];$$

if $\left[R \frac{\psi_1 - \psi_3}{2\Delta r} - \cos\theta\right] < 0$ and $\psi_2 - \psi_4 \geq 0$, then

(2.22c) $$L_3 M_0 = M_0 \left[\frac{-2r^2 \sin^2\theta}{(\Delta r)^2} - 2\frac{\sin^2\theta}{(\Delta\theta)^2} + \frac{1}{\Delta\theta}\left(R \sin\theta \frac{\psi_1 - \psi_3}{2\Delta r} - \sin\theta\cos\theta\right) - R\left(\frac{\psi_2 - \psi_4}{2\Delta\theta\Delta r}\right) \sin\theta\right] + M_1\left[\frac{r^2 \sin^2\theta}{(\Delta r)^2}\right] + M_2\left[\frac{\sin^2\theta}{(\Delta\theta)^2}\right] + M_3\left[\frac{r^2\sin^2\theta}{(\Delta r)^2} + R\left(\frac{\psi_2 - \psi_4}{2\Delta r\Delta\theta}\right)\sin\theta\right] + M_4\left[\frac{\sin^2\theta}{(\Delta\theta)^2} - \frac{1}{\Delta\theta}\left(R\sin\theta\frac{\psi_1 - \psi_3}{2\Delta r} - \sin\theta\cos\theta\right)\right];$$

if $\left[R \frac{\psi_1 - \psi_3}{2\Delta r} - \cos\theta\right] < 0$ and $\psi_2 - \psi_4 < 0$, then

(2.22d) $$L_3 M_0 = M_0 \left[\frac{-2r^2 \sin^2\theta}{(\Delta r)^2} - \frac{2\sin^2\theta}{(\Delta\theta)^2} + \frac{1}{\Delta\theta}\left(R \sin\theta \frac{\psi_1 - \psi_3}{2\Delta r} - \sin\theta\cos\theta\right) + \frac{R(\psi_2 - \psi_4)\sin\theta}{2\Delta\theta\Delta r}\right] + M_1\left[\frac{r^2 \sin^2\theta}{(\Delta r)^2}\right.$$

$$- R\left(\frac{\psi_2 - \psi_4}{2\Delta r \Delta \theta}\right) \sin \theta\Big] + M_2 \left[\frac{\sin^2 \theta}{(\Delta \theta)^2}\right] + M_3 \left[\frac{r^2 \sin^2 \theta}{(\Delta r)^2}\right]$$

$$+ M_4 \left[\frac{\sin^2 \theta}{(\Delta \theta)^2} - (R \sin \theta \; \frac{\psi_1 - \psi_3}{2\Delta r} - \sin \theta \cos \theta) \frac{1}{\Delta \theta}\right] .$$

The difference approximationsof (2.1) - (2.3) which will be used in the special way to be described in the next section are then

$$(2.23) \qquad L_1 \psi_0 = M_0 r^2 \sin \theta$$

$$(2.24) \qquad L_2 \Omega_0 = 0$$

$$(2.25) \qquad L_3 M_0 = R \left\{ \frac{\Omega_0}{r} \left[\left(\frac{\Omega_1 - \Omega_3}{\Delta r}\right) r \cos \theta - \left(\frac{\Omega_2 - \Omega_4}{\Delta \theta}\right) \sin \theta\right]\right.$$

$$\left. + \frac{M_0}{r}\left[\left(\frac{\psi_1 - \psi_3}{\Delta r}\right) r \cos \theta - \left(\frac{\psi_2 - \psi_4}{\Delta \theta}\right) \sin \theta\right]\right\} .$$


θ
2
(r, θ+Δθ)
(r−Δr, θ)
0
(r, θ)
(r+Δr, θ)
3
1
4
(r, θ−Δθ)
r

Fig. 2


θ
A*(r1, π/2+Δθ)
B*(r2, π/2+Δθ)
π/2+Δθ
π/2
ψ, Ω, ψr
ψ, Ω, ψr
D(r1, 0)
C(r2, 0)
r
ψ, Ω, M

Fig. 3

3. The Numerical Method

For fixed grid sizes Δr and $\Delta\theta$, let A^*, B^*, C, D be the points $(r_1, \frac{\pi}{2} + \Delta\theta)$, $(r_2, \frac{\pi}{2} + \Delta\theta)$, $(r_2, 0)$, $(r_1, 0)$, respectively, as shown in Figure 3. On and within rectangle A^*B^*CD, construct and number in the usual way [3] the set of interior grid points R_h and the set of boundary grid points S_h.

In general, we will aim at constructing on $R_h + S_h$ a triple sequence of discrete functions

$$(3.1) \qquad \psi^{(0)}, \psi^{(1)}, \psi^{(2)}, \psi^{(3)}, \ldots.$$

$$(3.2) \qquad \Omega^{(0)}, \Omega^{(1)}, \Omega^{(2)}, \Omega^{(3)}, \ldots.$$

$$(3.3) \qquad M^{(0)}, M^{(1)}, M^{(2)}, M^{(3)}, \ldots. ,$$

with the properties that, for some integral value k, and for given positive tolerances $\varepsilon_1, \varepsilon_2, \varepsilon_3$,

$$(3.4) \qquad |\psi^{(k)} - \psi^{(k+1)}| < \varepsilon_1$$

$$(3.5) \qquad |\Omega^{(k)} - \Omega^{(k+1)}| < \varepsilon_2$$

$$(3.6) \qquad |M^{(k)} - M^{(k+1)}| < \varepsilon_3 ,$$

uniformly on $R_h + S_h$. Each of the discrete functions in sequences (3.1) - (3.3) will be called an outer iterate. For $j = 1, 2, \ldots.$, each $\psi^{(j)}$ will be a solution of (2.23), each $\Omega^{(j)}$ will be a solution of

(2.24), and each $M^{(j)}$ will be a solution of a special form of (2.25). Numerical convergence to the tolerances given by (3.4) - (3.6) will yield the desired approximate solution $\psi^{(k+1)}$, $\Omega^{(k+1)}$, $M^{(k+1)}$.

Specifically, the algorithm proceeds as follows.

Step 1 Given Δr and $\Delta\theta$, construct R_h and S_h on rectangle A^*B^*CD.

Observe that by (1.8) - (1.10) and by the symmetry of the problem, one knows

$$\psi, \Omega, M, \qquad \text{on base } CD;$$
$$\psi, \Omega, \psi_r, \qquad \text{on } A^*D \text{ and } B^*C.$$

The problem then is to find M on the set theoretic difference $(R_h \cup S_h) - CD$, and to find ψ and Ω on the set theoretic difference $[R_h \cup S_h] - [(A^*D) + (DC) + (B^*C)]$. In the steps which follow, ψ, Ω and M are generated numerically only on those grid points where they are not already known.

Step 2. Set $\psi^{(0)} = 0$, $\Omega^{(0)} = 0$, $M^{(0)} = 0$.

Step 3. At each point of R_h of the form $(r_1 + \Delta r, j\Delta\theta)$, $j = 1, 2, \ldots, \frac{\pi}{2\Delta\theta}$, write down the equation

$$\psi^{(k)}(r_1 + \Delta r, j\Delta\theta) = \frac{\psi^{(k)}(r_1 + 2\Delta r, j\Delta\theta)}{4} . \tag{3.7}$$

At each point of R_h of the form $(r_2 - \Delta r, j\Delta\theta)$, $j = 1, 2, \ldots, \frac{\pi}{2\Delta\theta}$, write down the equation

$$(3.8) \qquad \psi^{(k)}(r_2 - \Delta r, j\Delta\theta) = \frac{\psi^{(k)}(r_2 - 2\Delta r, j\Delta\theta)}{4} .$$

At the remaining points of R_h, write down (2.23) in the form

$$(3.9) \qquad L_1\psi_0^{(k)} = M_0^{(k-1)} r^2 \sin\theta .$$

At each point of S_h of the form $(r_1 + j\Delta r, \frac{\pi}{2} + \Delta\theta)$, $j = 1, 2, \ldots, \frac{r_2 - \Delta r - r_1}{\Delta r}$, write down (3.9) with the symmetry substitution

$$(3.10) \qquad \psi^{(k)}(r_1 + j\Delta r, \frac{\pi}{2} + \Delta\theta) = -\psi^{(k)}(r_1 + j\Delta r, \frac{\pi}{2} - \Delta\theta).$$

Solve system (3.7) - (3.10) by SOR (successive overrelaxation) with overrelaxation factor r_ψ and convergence tolerance α_1. Call the solution $\bar{\psi}^{(k)}$ and define $\psi^{(k)}$ by the smoothing formula

$$(3.11) \qquad \psi^{(k)} = \rho\psi^{(k-1)} + (1 - \rho)\bar{\psi}^{(k)}, \quad 0 \leq \rho \leq 1 .$$

Step 4. At each point of R_h, write down (2.24) in the form

$$(3.12) \qquad L_2\Omega_0^{(k)} = 0$$

with the values of $\psi^{(k)}$ generated in Step 3 used for the coefficients of L_2. At each point of S_h of the form $(r_1 + j\Delta r, \frac{\pi}{2} + \Delta\theta)$, $j = 1, 2, \ldots, \frac{r_2 - \Delta r - r_1}{\Delta r}$, write down (3.12) with the symmetry substitution

$$(3.13)\qquad \Omega^{(k)}(r_1 + j\Delta r, \frac{\pi}{2} + \Delta\theta) = \Omega^{(k)}(r_1 + j\Delta r, \frac{\pi}{2} - \theta).$$

Solve system (3.12) - (3.13) by SOR with r_Ω and tolerance α_2. Call the solution $\bar{\Omega}^{(k)}$ and define $\Omega^{(k)}$ by

$$\Omega^{(k)} = \mu\Omega^{(k-1)} + (1-\mu)\bar{\Omega}^{(k)}, \quad 0 \le \mu \le 1.$$

Step 5. At each point of A^*D of the form $(r_1, j\Delta\theta)$, $j = 1, 2, \ldots, \frac{\pi}{2\Delta\theta}$, write down

$$(3.14)\qquad \bar{M}^{(k)} = \frac{2\psi^{(k)}(r_1 + \Delta r, j\Delta\theta)}{(\Delta r)^2}.$$

At each point of B^*C of the form $(r_2, j\Delta\theta)$, $j = 1, 2, \ldots, \frac{\pi}{2\Delta\theta}$, write down

$$(3.15)\qquad \bar{M}^{(k)} = \frac{2\psi^{(k)}(r_2 - \Delta r, j\Delta\theta)}{(\Delta r)^2}.$$

Define $M^{(k)}$ at the above points in A^*D and B^*C by smoothing (3.14) - (3.15) in the form

$$(3.16)\qquad M^{(k)} = \delta_1 M^{(k-1)} + (1-\delta_1)\bar{M}^{(k)}, \quad 0 \le \delta_1 \le 1.$$

Step 6. At each point of R_h, write down (2.25) in the form

$$(3.17)\qquad L_3 M_0^{(k)} = R\left\{\frac{\Omega_0^{(k)}}{r}\left[\left(\frac{\Omega_1^{(k)} - \Omega_3^{(k)}}{\Delta r}\right) r\cos\theta - \left(\frac{\Omega_2^{(k)} - \Omega_4^{(k)}}{\Delta\theta}\right)\sin\theta\right] + \frac{M_0^{(k-1)}}{r}\left[\left(\frac{\psi_1^{(k)} - \psi_3^{(k)}}{\Delta r}\right) r\cos\theta - \left(\frac{\psi_2^{(k)} - \psi_4^{(k)}}{\Delta\theta}\right)\sin\theta\right]\right\},$$

where the known values generated in Step 5 are inserted, and where the coefficients of L_3 are determined by using $\psi^{(k)}$. At each point of S_h of the form $(r_1 + j\Delta r, \frac{\pi}{2} + \Delta\theta)$, $j = 1, 2, \ldots, \frac{r_2 - \Delta r - r_1}{\Delta r}$, write down (3.17) with the symmetry substitution

$$M^{(k)}(r_1 + j\Delta r, \frac{\pi}{2} + \Delta\theta) = - M^{(k)}(r_1 + j\Delta r, \frac{\pi}{2} - \Delta\theta) \quad (3.18)$$

Solve (3.17) - (3.18) by SOR with r_M and tolerance α_3. Call the solution $\bar{M}^{(k)}$ and define $M^{(k)}$ on the same point set by

$$M^{(k)} = \delta_2 M^{(k-1)} + (1 - \delta_2) \bar{M}^{(k)}, \quad 0 \leq \delta_2 \leq 1.$$

<u>Step 7.</u> Do steps 3-6 for $k = 1, 2, \ldots,$ until (3.4) - (3.6) are valid. Check the residuals and read out $\psi^{(k+1)}$, $\Omega^{(k+1)}$, $M^{(k+1)}$ as the solution of the problem.

For a complete FORTRAN program of the method of this section, see Schubert [13].

4. Examples

Out of the large variety of examples which were run on the UNIVAC 1108 at the University of Wisconsin, Table 1 summarizes a typical cross section of problems with $r_1 = 0.5$ and $r_2 = 1$, whose solutions reflect variations of R, ω_1, and ω_2. As indicated in the table, graphs of level ψ curves are given in Figures 4-20. Refined grid calculations are presented in Figures 20 and 21. Convergence for the solutions shown in Figures 4-19 were each obtained in under 30 seconds, while the results shown in Figures 20-21 were obtained in under 7 minutes. For $R > 400$, convergence could not be obtained without modifying the numerical method as follows: in place of setting $\psi^{(0)} = \Omega^{(0)} = M^{(0)} = 0$ in Step 2, input the numerical solution obtained for $R = 400$.

Physically, for $\omega_1 = 0$ and $\omega_2 = 1$, the flows for $R = 10$ and $R = 100$ agree completely with those produced by Pearson [8]. However, as shown in Figures 6-11, the development of the recirculation zone near the equator as R increases is of a somewhat different character than that produced by Pearson in that various level curves for ψ need not be simply connected. Indeed, such a breakup is also evident in the central portions of Figures 7-11. The refined calculations for Figures 20-21 verified this phenomenon. Figures 12-17 show the variation in flow when ω_1 and ω_2 are approximately equal for various values of R. For $\omega_1 = 1$ and $\omega_2 = 0$, one should note the sharp flow concentrations in the equatorial region, as shown in Figure 18, and the relatively large increase in the values of ψ when ω_1 is increased to 10, as shown in Figure 19.

Finally, it should be noted that the above results should be considered to be only qualitative. Calculations with much finer grids are necessary for highly reliable quantitative results. Such refined calculations should also reveal the degree of sensitivity of the method near the singular line $\theta = 0$. If the method is ultrasensitive near this line, then it need only be modified by transposing all initial data given on $\theta = 0$ to $\theta = \delta$, where $\delta > 0$, and then by applying the method on the resulting truncated region. Our limited calculations do not indicate such an ultrasensitivity.

TABLE 1

ω_1	ω_2	Δr	$\Delta\theta$	R	ε	μ	δ_1	δ_2	r_ψ	r_Ω	r_M	α_1	α_2	α_3	ε_1	ε_2	ε_3	See Fig
0	1	.05	$\frac{\pi}{40}$	10	0.1	0.1	0.9	0.1	1.8	1.3	1.3	$5\cdot10^{-7}$	$2.5\cdot10^{-4}$	$5\cdot10^{-5}$	$4\cdot10^{-7}$	10^{-4}	10^{-4}	4
0	1	.05	$\frac{\pi}{40}$	100	0.1	0.1	0.1	0.9	1.8	1.3	1.3	$5\cdot10^{-7}$	$2.5\cdot10^{-4}$	$5\cdot10^{-5}$	3 $5\cdot10^{-6}$	10^{-4}	$2\cdot10^{-3}$	5
0	1	.05	$\frac{\pi}{40}$	400	0.1	0.1	0.1	0.9	1.8	1.2	1.2	$5\cdot10^{-7}$	$2.5\cdot10^{-4}$	$5\cdot10^{-5}$	10^{-5}	10^{-4}	$3\cdot10^{-4}$	6
0	1	.05	$\frac{\pi}{40}$	1000	0.05	0.05	0.05	0.98	1.8	1.1	1.1	$5\cdot10^{-7}$	$5\cdot10^{-5}$	$5\cdot10^{-5}$	10^{-5}	$5\cdot10^{-4}$	$8\cdot10^{-3}$	7
0	1	.05	$\frac{\pi}{40}$	2000	0.05	0.05	0.05	0.9995	1.8	1.1	1.1	$5\cdot10^{-7}$	$5\cdot10^{-5}$	$5\cdot10^{-5}$	10^{-5}	$5\cdot10^{-4}$	$8\cdot10^{-3}$	8
0	1	.05	$\frac{\pi}{40}$	3000	0.05	0.05	0.05	0.9995	1.8	1.1	1.1	$5\cdot10^{-7}$	10^{-4}	$5\cdot10^{-5}$	10^{-5}	$5\cdot10^{-4}$	$8\cdot10^{-3}$	9
0	1	.05	$\frac{\pi}{40}$	5000	0.05	0.05	0.05	0.995	1.8	1.1	1.1	$5\cdot10^{-7}$	$5\cdot10^{-5}$	$5\cdot10^{-5}$	10^{-5}	$5\cdot10^{-4}$	$3\cdot10^{-3}$	10
0	1	.05	$\frac{\pi}{40}$	10000	0.05	0.05	0.05	0.9995	1.8	1.1	1.1	$5\cdot10^{-7}$	10^{-4}	$5\cdot10^{-5}$	$5\cdot10^{-5}$	$5\cdot10^{-4}$	$4\cdot10^{-3}$	11
1	1.2	.05	$\frac{\pi}{40}$	100	.1	.1	.1	.9	1.5	1.1	1.4	$5\cdot10^{-7}$	10^{-4}	$5\cdot10^{-5}$	$5\cdot10^{-8}$	10^{-5}	$5\cdot10^{-6}$	12
1	1.1	.05	$\frac{\pi}{40}$	100	.1	.1	.1	.9	1.5	1.1	1.4	$5\cdot10^{-7}$	10^{-4}	$5\cdot10^{-5}$	$5\cdot10^{-8}$	10^{-5}	$5\cdot10^{-6}$	13
1	1.01	.05	$\frac{\pi}{40}$	100	.1	.1	.1	.9	1.5	1.1	1.4	$5\cdot10^{-7}$	10^{-4}	$5\cdot10^{-5}$	$5\cdot10^{-8}$	10^{-5}	$5\cdot10^{-6}$	14
1	1.2	.05	$\frac{\pi}{40}$	400	.1	.1	.1	.95	1.8	1.1	1.1	$5\cdot10^{-7}$	10^{-4}	$5\cdot10^{-5}$	10^{-6}	$5\cdot10^{-4}$	$5\cdot10^{-4}$	15
1	1.02	.05	$\frac{\pi}{40}$	2500	0.05	0.05	0.05	0.9995	1.5	1.1	1.4	$5\cdot10^{-7}$	10^{-4}	$5\cdot10^{-5}$	10^{-4}	$5\cdot10^{-4}$	$5\cdot10^{-4}$	16
1	1.01	.05	$\frac{\pi}{40}$	10000	0.05	0.05	0.05	0.99999	1.5	1.1	1.5	10^{-7}	$5\cdot10^{-5}$	$2.5\cdot10^{-5}$	10^{-7}	$5\cdot10^{-4}$	$5\cdot10^{-4}$	17
1	0	.05	$\frac{\pi}{40}$	100	0.1	0.1	0.9	0.1	1.8	1.3	1.3	$5\cdot10^{-7}$	$2.5\cdot10^{-4}$	$5\cdot10^{-5}$	10^{-5}	10^{-4}	10^{-4}	18
10	0	.05	$\frac{\pi}{40}$	100	0.9	0.1	0.1	0.1	1.8	1.3	1.3	$5\cdot10^{-7}$	$2.5\cdot10^{-4}$	$5\cdot10^{-5}$	10^{-4}	$2\cdot10^{-4}$	$3\cdot10^{-3}$	19
0	1	.025	$\frac{\pi}{80}$	3000	0.05	0.05	0.05	0.9995	1.8	1.1	1.1	10^{-7}	10^{-5}	10^{-5}	$5\cdot10^{-6}$	10^{-3}	$3\cdot10^{-3}$	20,21



Fig. 4
Level ψ curves, multiplied by 10^4, for R=10, $w_1 = 0$, $w_2 = 1$.



Fig. 5
Level ψ curves, multiplied by 10^4, for R=100, $w_1 = 0$, $w_2 = 1$.



Fig. 6
Level ψ curves, multiplied by 10^4, for R=400, $w_1 = 0$, $w_2 = 1$.



Fig. 7
Level ψ curves, multiplied by 10^4, for R=1000, $w_1 = 0$, $w_2 = 1$



Fig. 8
Level ψ curves, multiplied by 10^4, for R=2000, $w_1 = 0$, $w_2 = 1$.



Fig. 9
Level ψ curves, multiplied by 10^4, for R=3000, $w_1=0$, $w_2=1$.



Fig. 10
Level ψ curves, multiplied by 10^4, for R=5000, $w_1 = 0$, $w_2 = 1$.



Fig. 11
Level ψ curves, multiplied by 10^4, for R=10000, $w_1=0$, $w_2=1$.



Fig. 12
Level ψ curves, multiplied by 10^4, for $R = 100$, $w_1 = 1$, $w_2 = 1.2$.



Fig. 13
Level ψ curves, multiplied by 10^4, for $R = 100$, $w_1 = 1$, $w_2 = 1.1$.



Fig. 14
Level ψ curves, multiplied by 10^4, for $R = 100$, $w_1 = 1$, $w_2 = 1.01$.



Fig. 15
Level ψ curves, multiplied by 10^4, for $R = 400$, $w_1 = 1$, $w_2 = 1.2$.



Fig. 16
Level ψ curves, multiplied by 10^4, for $R = 2500, \varpi_1 = 1, \varpi_2 = 1.02$.



Fig. 17
Level ψ curves, multiplied by 10^4, for $R = 10000, \varpi_1 = 1, \varpi_2 = 1.01$.



Fig. 18
Level ψ curves, multiplied by 10^4, for $R=100, \varpi_1 = 1, \varpi_2 = 0$.



Fig. 19
Level ψ curves, multiplied by 10^4, for $R=100, \varpi_1 = 10, \varpi_2 = 0$.



Fig. 20
Level ψ curves, multiplied by 10^4, for $R = 3000$, $\omega_1 = 0$, $\omega_2 = 1$.



Fig. 21
Level Ω curves for $R = 3000$, $\omega_1 = 0$, $\omega_2 = 1$.

REFERENCES

1. P. L. Bhatnagar and G. K. Rajeswari, "Secondary flow of non-Newtonian fluids between two concentric spheres rotating about an axis," Indian J. Math., 5, 1963, p. 93.

2. Ju. K. Bratuhin, "On the evaluation of the critical Reynolds number for the flow of fluid between two rotating spherical surfaces," Prikl. Mat. Meh., 25, 1961, p. 858; translated in J. Appl. Math. Mech., 25, 1962, p. 1286.

3. D. Greenspan, Lectures on the Numerical Solution of Linear, Singular and Nonlinear Differential Equations, Prentice Hall, Englewood Cliffs, N. J., 1968.

4. H. P. Greenspan, The Theory of Rotating Fluids, Cambridge Univ. Press, Cambridge, 1968.

5. W. L. Haberman, "Secondary flow about a sphere rotating in a viscous liquid inside a coaxially rotating spherical container," Phys. Fluids, 5, 1962, p. 625; erratum, ibid, 5, 1962, p. 1136.

6. W. E. Langlois, "Steady flow of a slightly viscoelastic fluid between rotating spheres," Quart. Appl. Math., 2, 1963, p. 61.

7. Ju. G. Ovseenko, "On the motion of a viscous fluid between two rotating spheres," Izv. Vyss. Ucebn. Zaved. Mat., 1963, 35, p. 129.

8. C. E. Pearson, "A numerical study of time-dependent viscous flow between two rotating spheres," J. Fluid Mech., 28, 1967, p. 323.

9. C. E. Pearson, "A numerical method for incompressible, viscous flow problems in a spherical geometry," in Studies in Numerical Analysis I, SIAM, 1969, p. 65.

10. J. Pedlosky, "Axially symmetric motion of a stratified, rotating fluid in a spherical annulus of narrow gap," J. Fluid Mech., 36, 1969, p. 401.

11. I. Proudman, "The almost rigid rotation of a viscous fluid between concentric spheres," J. Fluid Mech., 1, 1956, p. 505.

12. A. Schubert, "Fortran Program for Flow between Rotating Spheres - Figure 12 Case", Appendix, TR 120, Department of Computer Sciences, Univ. of Wisconsin, Madison, Wisconsin, 1971.

13. K. Sterwartson, "On almost rigid rotations. Part 2," J. Fluid Mech., 26, 1966, p. 131.

APPENDIX

A. Schubert's Fortran Program for Flow between Rotating Spheres

```
C
      PARAMETER NRMAX=21,NTHMAX=41
C
      REAL M(NRMAX,NTHMAX,4),MU,MU1
C
      COMMON/BA/B(NRMAX,NTHMAX,4),AA(NRMAX,NTHMAX)/NR/NR/NTH/NTH
     */NRTHM/NRM1,NRM2,NTHM1,NTHM2/REL/REL(3)/BLOC1/RSQD(NRMAX),
     * STH(NTHMAX),CTHD(NTHMAX),STHD(NTHMAX),R(NRMAX),COTTH(NTHMAX)
     * /FUN/PSI(NRMAX,NTHMAX,4),OME(NRMAX,NTHMAX,4),M/RE/RE/BLOC2/DRSQ,
     * DR2,DTH2/IVAR/IVAR
C
      DIMENSION RAD(2),W(2),DELTA(2),EPS(3),CTH(NTHMAX),C(NRMAX,NTHMAX),
     * NSOR(3),ALPHA(3),ANGVEL(NRMAX,NTHMAX)
C
      DATA HALFPI/1.5707963/,RDCONV/0.017453292/,MAXSOR,MAXOUT/
     * 1000,75/
      DATA IBUG/0/,IBEGIN/81/
C
   99 FORMAT(12E5.5)
   98 FORMAT('0TERMINATED FOR MAX. OUTER ITERATIONS')
   97 FORMAT('0MAX. SOR ITERATIONS FOR PSI')
   96 FORMAT('0MAX. SOR ITERATIONS FOR OME')
   95 FORMAT('0MAX. SOR ITERATIONS FOR M')
   94 FORMAT('0CONVERGENCE. TOTAL SOR ITERATIONS =' 3I/)
   93 FORMAT('1R(1)=' F4.1,3X 'R(2)=' F4.1,3X 'OMEGA(1)=' F4.1,3X
     * 'OMEGA(2)=' F4.1,3X 'DR=' F6.4,3X 'DTHETA=' F7.3,3X 'REYNOLDS NO.
     * =' F8.0)
   92 FORMAT(1H0 8X 'SMOOTHING' 9X 'RELAX' 8X 'SOR TOL.' 4X 'OUTER TOL.'
     * /1H0 2X 'PSI' F8.4,13X F5.3,2E14.2/1X 'OMEGA' F8.4,13X F5.3,
     * 2E14.2/5X 1HM F8.4,F7.4,F11.3,2E14.2)
   91 FORMAT('0OUTER ITER.' I5,5X 'PSI * 10**3' 5X 'SOR ITERATIONS ='I5)
   90 FORMAT('0INITIAL VALUES TAKEN FROM OUTPUT FROM  EARLIER CASE')
   89 FORMAT('0INITIAL VALUES'/)
   88 FORMAT(19X ' OMEGA * 10**0' 5X 'SOR ITERATIONS =' I5)
   87 FORMAT(23X ' M * 10**0' 5X 'SOR ITERATIONS =' I5)
   86 FORMAT(1X 11E11.6)
   85 FORMAT(' SOR ITER.' I5)
   84 FORMAT('0COEFFICIENTS IN SOR FOR OME')
   83 FORMAT('0COEFFICIENTS IN SOR FOR PSI')
   82 FORMAT('0COEFFICIENTS IN SOR FOR M')
   81 FORMAT('0ANGULAR VELOCITY'/)
C
C     READ RADII AND ANGULAR VELOCITIES OF INNER AND OUTER SPHERES,
C     RADIAL AND ANGULAR INCREMENTS FOR DISCRETE APPROXIMATION,
C     AND REYNOLDS NO.
C
      INVAL=0
    5 READ(5,99) RAD,W,DR,DTHDG,RE,FINI
C
C     DETERMINE NO. OF GRID POINTS IN EACH OF RADIAL AND ANGULAR
C     DIRECTIONS.
C
```

```
      NR=(RAD(2)-RAD(1))/DR+1.05
      DTH=DTHDG*RDCONV
      NTH=90./DTHDG+1.05
      NRM1=NR-1
      NRM2=NR-2
      NTHM1=NTH-1
      NTHM2=NTH-2
C
C     COMPUTE AND STORE AWAY ONCE AND FOR ALL VARIOUS CONSTANTS TO BE
C     USED IN PROGRAM BELOW.
C
      DR2=2.*DR
      DRSQ=DR*DR
      DTH2=2.*DTH
      DRDTH2=DR*DTH2
      DTHSQ=DTH**2
      DO 6 I=1,NR
      R(I)=RAD(1)+FLOAT(I-1)*DR
    6 RSQD(I)=R(I)**2/DRSQ
      DO 7 J=2,NTH
      THETA=FLOAT(J-1)*DTH
      STH(J)=SIN(THETA)
      CTH(J)=COS(THETA)
      IF(J.EQ.NTH   ) CTH(J)=0.
      STHD(J)=STH(J)/DTHSQ
      CTHD(J)=CTH(J)/DTH
    7 COTTH(J)=CTH(J)/STH(J)
      IF(IBUG.EQ.0) GO TO 107
      WRITE(6,86) (STH(J),J=1,NTH),(CTH(J),J=1,NTH),(STHD(J),J=1,NTH),
     * (CTHD(J),J=1,NTH),(COTTH(J),J=1,NTH)
C
C     SET BOUNDARY CONDITIONS
C
C     PSI AND OMEGA ON BOUNDARIES R=R1 AND R=R2
C
  107 DO 8 J=2,NTH
      PSI(1,J,3)=0.
      PSI(NR,J,3)=0.
      OME(1,J,3)=W(1)*(RAD(1)*STH(J))**2
    8 OME(NR,J,3)=W(2)*(RAD(2)*STH(J))**2
C
C     PSI,OMEGA,M ON BOUNDARY THETA=0
C
      DO 9 I=2,NRM1
      PSI(I,1,3)=0.
      OME(I,1,3)=0.
    9 M(I,1,3)=0.
C
C     SET BOUNDARY CONDITIONS IN OTHER WORK ARRAYS FOR PSI,OMEGA,M
C
      DO 11 K=1,2
      DO 10 J=2,NTH
      PSI(1,J,K)=PSI(1,J,3)
      PSI(NR,J,K)=PSI(NR,J,3)
      OME(1,J,K)=OME(1,J,3)
   10 OME(NR,J,K)=OME(NR,J,3)
      DO 11 I=2,NRM1
```

```
      PSI(I,1,K)=PSI(I,1,3)
      OME(I,1,K)=OME(I,1,3)
   11 M(I,1,K)=M(I,1,3)
C
C     READ CONVERGENCE TOLERANCES AND SMOOTHING AND OVER-RELAXATION
C     FACTORS.
C
   12 READ(5,99) EPS,ALPHA,FIN2
   13 READ(5,99) RHO,MU,DELTA,REL,SAVE,USE,FIN3
      WRITE(6,93) RAU,,DR,DTHDG,RE
      WRITE(6,92) RHO,REL(1),ALPHA(1),EPS(1),MU,REL(2),ALPHA(2),EPS(2),
     * DELTA,REL(3),ALPHA(3),EPS(3)
      R14=.25*REL(1)
      R32=REL(3)/REL(2)
      RHO1=1.-RHO
      MU1=1.-MU
      DEL11=1.-DELTA(1)
      DEL21=1.-DELTA(2)
      IF(.NOT.(USE.GT.0.)) GO TO 513
      IF(USE.LT. 1.5) GO TO 1513
      DO 1013 I=1,NR,2
      II=I/2+1
      DO 1013 J=1,NTH,2
      JJ=J/2+1
      PSI(I,J,3)=PSI(II,JJ,4)
      OME(I,J,3)=OME(II,JJ,4)
 1013   M(I,J,3)=  M(II,JJ,4)
      DO 1113 I=1,NR,2
      DO 1113 J=2,NTHM1,2
      PSI(I,J,3)=.5*(PSI(I,J-1,3)+PSI(I,J+1,3))
      OME(I,J,3)=.5*(OME(I,J-1,3)+OME(I,J+1,3))
 1113   M(I,J,3)=.5*(  M(I,J-1,3)+  M(I,J+1,3))
      DO 1213 J=1,NTH,2
      DO 1213 I=2,NRM1,2
      PSI(I,J,3)=.5*(PSI(I-1,J,3)+PSI(I+1,J,3))
      OME(I,J,3)=.5*(OME(I-1,J,3)+OME(I+1,J,3))
 1213   M(I,J,3)=.5*(  M(I-1,J,3)+  M(I+1,J,3))
      DO 1313 I=2,NRM1,2
      DO 1313 J=2,NTHM1,2
      PSI(I,J,3)=.25*(PSI(I-1,J,3)+PSI(I+1,J,3)+PSI(I,J-1,3)+PSI(I,J+1,3
     *))
      OME(I,J,3)=.25*(OME(I-1,J,3)+OME(I+1,J,3)+OME(I,J-1,3)+OME(I,J+1,3
     *))
 1313   M(I,J,3)=.25*(  M(I-1,J,3)+  M(I+1,J,3)+  M(I,J-1,3)+  M(I,J+1,3
     *))
      GO TO 313
 1513 DO 113 J=1,NTH
      DO 113 I=1,NR
      PSI(I,J,3)=PSI(I,J,4)
      OME(I,J,3)=OME(I,J,4)
  113 M(I,J,3)=M(I,J,4)
  313 WRITE(6,90)
      IF(INVAL.EQ.0) GO TO 1114
      INVAL=0
      WRITE(6,89)
      IVAR=1
      CALL OUTPUT(PSI(1,1,3))
```

```
      IVAR=2
      CALL OUTPUT(OME(1,1,3))
      IVAR=3
      CALL OUTPUT(M(1,1,3))
      GO TO 1114
C
C     INITIALIZE ALL FUNCTIONS FOR FIRST OUTER ITERATION.
C
 513  DO 14 J=2,NTH
      DO 14 I=2,NRM1
      PSI(I,J,3)=0.
      OME(I,J,3)=0.
   14 M(I,J,3)=0.
      DO 1014 K=1,3
      DO 1014 J=2,NTH
      M(1,J,K)=0.
 1014 M(NR,J,K)=0.
C
 1114 DO 114 I=1,3
  114 NSOR(I)=0
      IOUT=0
   15 IOUT=IOUT+1
C
C     UPDATE ALL FUNCTIONS FOR NEXT OUTER ITERATION
C
C     UPDATE ALL FUNCTIONS IN INTERIOR.
C
      DO 16 I=2,NRM1
      DO 16 J=2,NTH
      PSI(I,J,1)=PSI(I,J,3)
      OME(I,J,1)=OME(I,J,3)
   16 M(I,J,1)=M(I,J,3)
C
C     THEN M ON BOUNDARIES R=R1 AND R=R2
C
      DO 116 J=2,NTH
      M(1,J,1)=M(1,J,3)
  116 M(NR,J,1)=M(NR,J,3)
C
C     DETERMINE PSI AT CURRENT OUTER ITERATION BY SOR
C
C     FIRST COMPUTE COEFFICIENTS
C
      B0=1.-REL(1)
      DO 117 J=2,NTH
      TMP=STHD(J)+CTHD(J)
      B(2,J,1)=R14
      B(2,J,2)=0.
      B(2,J,3)=0.
      B(2,J,4)=0.
      C(2,J)=0.
      DO 17 I=3,NRM2
      A1=RSQD(I)*STH(J)
      A0=-2.*A1-2.*STHD(J)-CTHD(J)
      RA0=-REL(1)/A0
      B(I,J,1)=A1*RA0
      B(I,J,2)=STHD(J)*RA0
```

```
      B(I,J,3)=B(I,J,1)
      B(I,J,4)=TMP*RAD
   17 C(I,J)=M(I,J,1)*DRSQ*A1*RAD
      B(NRM1,J,1)=0.
      B(NRM1,J,2)=0.
      B(NRM1,J,3)=R14
      B(NRM1,J,4)=0.
  117 C(NRM1,J)=0.
      IF(IBUG.EQ.0) GO TO 1017
      WRITE(6,83)
      DO 517 K=1,4
  517 WRITE(6,86) ((B(NR-I+1,J,K),J=1,NTH),I=1,NR)
      WRITE(6,86) ((C(NR-I+1,J),J=1,NTH),I=1,NR)
C
C     THEN ITERATE
C
 1017 IPSI=0
   18 IF(IPSI.GE.MAXSOR) GO TO 61
      IPSI=IPSI+1
C
C     UPDATE *PREVIOUS VALUE* VECTOR FOR PSI.
C
      DO 19 J=2,NTH
      DO 19 I=2,NRM1
   19 PSI(I,J,2)=PSI(I,J,3)
C
C     ITERATION EQUATION FOR PSI IN INTERIOR
C
      DO 21 J=2,NTHM1
      DO 21 I=2,NRM1
   21 PSI(I,J,3)=B0*PSI(I,J,2)+B(I,J,1)*PSI(I+1,J,2)+B(I,J,2)*
     * PSI(I,J+1,2)+B(I,J,3)*PSI(I-1,J,3)+B(I,J,4)*PSI(I,J-1,3)-C(I,J)
C
C     ITERATION EQUATION FOR PSI ON THE LINE THETA=PI/2
C
      PSI(2,NTH,3)=B0*PSI(2,NTH,2)+R14*PSI(3,NTH,2)
      DO 23 I=3,NRM2
   23 PSI(I,NTH,3)=B0*PSI(I,NTH,2)+B(I,NTH,1)*PSI(I+1,NTH,2)-B(I,NTH,2)
     $ *PSI(I,NTHM1,3)+B(I,NTH,3)*PSI(I-1,NTH,3)+B(I,NTH,4)*PSI(I,NTHM1,
     * 3)-C(I,NTH)
      PSI(NRM1,NTH,3)=B0*PSI(NRM1,NTH,2)+R14*PSI(NRM2,NTH,3)
      IF(IBUG.EQ.0) GO TO 123
      WRITE(6,85) IPSI
      WRITE(6,86) ((PSI(NR-I+1,J,3),J=1,NTH),I=1,NR)
C
C     TEST FOR CONVERGENCE OF SOR ITERATION FOR PSI
C
  123 DO 24 J=2,NTH
      DO 24 I=2,NRM1
      IF(.NOT.(ABS(PSI(I,J,3)-PSI(I,J,2)).LT.ALPHA(1))) GO TO 18
   24 CONTINUE
C
C     CONVERGENCE ATTAINED. SMOOTH SOLUTION.
C
      DO 25 J=2,NTH
      DO 25 I=2,NRM1
   25 PSI(I,J,3)=RHO*PSI(I,J,1)+RHO1*PSI(I,J,3)
```

```
C
C     OUTPUT SMOOTHED SOLUTION
C
      IF(SAVE.GT.0.) GO TO 26
      IF(IOUT.LT.IBEGIN) GO TO 26
      WRITE(6,91) IOUT,IPSI
      IVAR=1
      CALL OUTPUT(PSI(1,1,3))
   26 NSOR(1)=NSOR(1)+IPSI
C
C     NEXT COMPUTE COEFFICIENTS FOR DETERMINING OMEGA BY SOR.
C
      B0=1.-REL(2)
      DO 28 J=2,NTH
      DO 28 I=2,NRM1
      T1=RSQD(I)*STH(J)
      T2=(RE*(PSI(I+1,J,3)-PSI(I-1,J,3))/DR2-CTH(J))/DTH
      IF(J.LT.NTH) GO TO 27
      T3=0.
      GO TO 127
   27 T3=RE*(PSI(I,J+1,3)-PSI(I,J-1,3))/DRDTH2
  127 AA(I,J)=-2.*T1-2.*STHD(J)-ABS(T2)-ABS(T3)
      RAA=-REL(2)/AA(I,J)
      B(I,J,1)=(T1-AMIN1(0.,T3))*RAA
      B(I,J,2)=(STHD(J)+AMAX1(0.,T2))*RAA
      B(I,J,3)=(T1+AMAX1(0.,T3))*RAA
   28 B(I,J,4)=(STHD(J)-AMIN1(0.,T2))*RAA
      IF(IBUG.EQ.0) GO TO 129
      WRITE(6,84)
      DO 29 K=1,4
   29 WRITE(6,86) ((B(NR-I+1,J,K),J=1,NTH),I=1,NR)
      WRITE(6,86) ((AA(NR-I+1,J),J=1,NTH),I=1,NR)
C
  129 IOME=0
   30 IF(IOME.GE.MAXSOR) GO TO 52
      IOME=IOME+1
C
C     UPDATE 'PREVIOUS VALUE' VECTOR FOR OMEGA
C
      DO 31 J=2,NTH
      DO 31 I=2,NRM1
   31 OME(I,J,2)=OME(I,J,3)
C
C     ITERATION EQUATION FOR OMEGA IN INTERIOR
C
      DO 32 J=2,NTHM1
      DO 32 I=2,NRM1
   32 OME(I,J,3)=B0*OME(I,J,2)+B(I,J,1)*OME(I+1,J,2)+B(I,J,2)*OME(I,J+1,
     * 2)+B(I,J,3)*OME(I-1,J,3)+B(I,J,4)*OME(I,J-1,3)
C
C     ITERATION EQUATION FOR OMEGA ON THE LINE THETA=PI/2
C
      DO 33 I=2,NRM1
   33 OME(I,NTH,3)=B0*OME(I,NTH,2)+B(I,NTH,1)*OME(I+1,NTH,2)+B(I,NTH,2)
     $ *OME(I,NTHM1,3)+B(I,NTH,3)*OME(I-1,NTH,3)+B(I,NTH,4)*OME(I,NTHM1,
     * 3)
      IF(IBUG.EQ.0) GO TO 133
```

```
      WRITE(6,85) IOME
      WRITE(6,86) ((OME(NR-I+1,J,3),J=1,NTH),I=1,NR)
C
C     TEST FOR CONVERGENCE OF SOR ITERATION FOR OMEGA
C
  133 DO 34 J=2,NTH
      DO 34 I=2,NRM1
      IF(.NOT.(ABS(OME(I,J,3)-OME(I,J,2)).LT.ALPHA(2))) GO TO 30
   34 CONTINUE
C
C     CONVERGENCE ATTAINED. SMOOTH AND OUTPUT SOLUTION.
C
      DO 35 J=2,NTH
      DO 35 I=2,NRM1
   35 OME(I,J,3)=MU*OME(I,J,1)+MU1*OME(I,J,3)
      IF(SAVE.GT.0.) GO TO 36
      IF(IOUT.LT.IBEGIN) GO TO 36
      WRITE(6,88) IOME
      IVAR=2
      CALL OUTPUT(OME(1,1,3))
   36 NSOR(2)=NSOR(2)+IOME
C
C     COMPUTE AND SMOOTH M ON BOUNDARIES R=R1 AND R=R2
C
      DO 38 J=2,NTH
      M(1,J,3)=DELTA(1)*M(1,J,1)+DEL11*2.*PSI(2,J,3)/DRSQ
   38 M(NR,J,3)=DELTA(1)*M(NR,J,1)+DEL11*2.*PSI(NRM1,J,3)/DRSQ
C
C     COMPUTE M IN INTERIOR - NOTE COEFFICIENTS ARE SAME AS FOR OME
C     EXCEPT FOR RHS (CONSTANT) TERM OF LINEAR SYSTEM.
C
      B0=1.-REL(3)
      DO 40 J=2,NTHM1
      DO 40 I=2,NRM1
      C(I,J)=RE*(2.*OME(I,J,3)*((OME(I+1,J,3)-OME(I-1,J,3))/DR2*COTTH(J)
     * -(OME(I,J+1,3)-OME(I,J-1,3))/(R(I)*DTH2))+2.*M(I,J,1)*
     * ((PSI(I+1,J,3)-PSI(I-1,J,3))/DR2*COTTH(J)-(PSI(I,J+1,3)-
     * PSI(I,J-1,3))/(R(I)*DTH2)))/AA(I,J)*(-REL(3))
C
C     ADJUST COEFFICIENTS FOR OVER-RELAXATION FACTOR FOR M INSTEAD
C     OF THAT FOR OMEGA BY MULTIPLYING BY R32=REL(3)/REL(2)
C
      DO 40 K=1,4
   40 B(I,J,K)=B(I,J,K)*R32
      IF(IBUG.EQ.0) GO TO 1140
      WRITE(6,82)
      DO 1040 K=1,4
 1040 WRITE(6,86) ((B(NR-I+1,J,K),J=1,NTH),I=1,NR)
      WRITE(6,86) ((C(NR-I+1,J),J=1,NTH),I=1,NR)
C
C     UPDATE 'PREVIOUS VALUE' VECTOR FOR M ON BOUNDARY R=R2
C
 1140 DO 140 J=2,NTH
  140 M(NR,J,2)=M(NR,J,3)
C
      IM=0
   41 IF(IM.GE.MAXSOR) GO TO 53
```

```
      IM=IM+1
C
C     UPDATE 'PREVIOUS VALUE' VECTOR FOR M IN INTERIOR
C
      DO 141 J=2,NTH
      DO 141 I=2,NRM1
  141 M(I,J,2)=M(I,J,3)
C
C     ITERATION EQUATION FOR M IN INTERIOR
C
      DO 42 J=2,NTHM1
      DO 42 I=2,NRM1
   42 M(I,J,3)=BO*M(I,J,2)+B(I,J,1)*M(I+1,J,2)+B(I,J,2)*M(I,J+1,2)
     * +B(I,J,3)*M(I-1,J,3)+B(I,J,4)*M(I,J-1,3)-C(I,J)
C
C     ITERATION EQUATION FOR M ON THE LINE THETA=PI/2
C
      DO 43 I=2,NRM1
   43 M(I,NTH,3)=BO*M(I,NTH,2)+B(I,NTH,1)*M(I+1,NTH,2)-B(I,NTH,2)*M(I,NT
     * HM1,3)+B(I,NTH,3)*M(I-1,NTH,3)+B(I,NTH,4)*M(I,NTHM1,3)
      IF(IBUG.EQ.0) GO TO 143
      WRITE(6,85) IM
      WRITE(6,86) ((M(NR-I+1,J,3),J=1,NTH),I=1,NR)
C
C     TEST FOR CONVERGENCE OF SOR ITERATION FOR M
C
  143 DO 44 J=2,NTH
      DO 44 I=2,NRM1
      IF(.NOT.(ABS(M(I,J,3)-M(I,J,2)).LT.ALPHA(3))) GO TO 41
   44 CONTINUE
C
C     CONVERGENCE ATTAINED. SMOOTH AND OUTPUT SOLUTION.
C
      DO 45 J=2,NTH
      DO 45 I=2,NRM1
   45 M(I,J,3)=DELTA(2)*M(I,J,1)+DEL21*M(I,J,3)
      IF(SAVE.GT.0.) GO TO 145
      IF(IOUT.LT.IBEGIN) GO TO 145
      WRITE(6,87) IM
      IVAR=3
      CALL OUTPUT(M(1,1,3))
  145 NSOR(3)=NSOR(3)+IM
C
C     TEST FOR CONVERGENCE OF OUTER ITERATION.
C
C     FIRST TEST M ON BOUNDARIES R=R1 AND R=R2
C
      DO 46 J=2,NTH
      IF(.NOT.(ABS(M(1,J,3)-M(1,J,1)).LT.EPS(3))) GO TO 50
      IF(.NOT.(ABS(M(NR,J,3)-M(NR,J,1)).LT.EPS(3))) GO TO 50
   46 CONTINUE
C
C     THEN TEST ALL FUNCTIONS IN INTERIOR
C
      DO 47 J=2,NTH
      DO 47 I=2,NRM1
      IF(.NOT.(ABS(M(I,J,3)-M(I,J,1)).LT.EPS(3))) GO TO 50
```

```
      IF(.NOT.(ABS(OME(I,J,3)-OME(I,J,1)).LT.EPS(2))) GO TO 50
      IF(.NOT.(ABS(PSI(I,J,3)-PSI(I,J,1)).LT.EPS(1))) GO TO 50
   47 CONTINUE
C
C     CONVERGENCE ATTAINED. COMPUTE RESIDUALS.
C
   48 WRITE(6,94) NSOR
      CALL RESID
      GO TO 54
   50 IF(IOUT.LT.MAXOUT) GO TO 15
      WRITE(6,98)
      IF(SAVE.GT. 0.) GO TO 154
      DO 150 J=2,NTH
      DO 150 I=2,NRM1
  150 ANGVEL(I,J)=OME(I,J,3)/(R(I)*STH(J))**2
      DO 250 J=2,NTH
      ANGVEL(1,J)=W(1)
  250 ANGVEL(NR,J)=W(2)
      WRITE(6,81)
      IVAR=4
      CALL OUTPUT(ANGVEL)
      GO TO 48
C
C     SOR CONVERGENCE FAILURE MESSAGES
C
   51 WRITE(6,97)
      GO TO 56
   52 WRITE(6,96)
      GO TO 56
   53 WRITE(6,95)
      GO TO 56
C
C     TEST FOR END OF INPUT AT EACH REPETITION LEVEL.
C
   54 IF(.NOT.(SAVE.GT.0.)) GO TO 56
  154 DO 55 J=1,NTH
      DO 55 I=1,NR
      PSI(I,J,4)=PSI(I,J,3)
      OME(I,J,4)=OME(I,J,3)
   55 M(I,J,4)=M(I,J,3)
      INVAL=1
   56 IF(ABS(FIN3).LT. .01) GO TO 13
      IF(ABS(FIN2).LT. .01) GO TO 12
      IF(ABS(FIN1).LT. .01) GO TO 5
      STOP
      END
```